# $Id: PKGBUILD 61068 2011-12-21 20:50:13Z andrea $
# Maintainer: Daniel J Griffiths <ghost1227@archlinux.us>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Wes Brewer <brewerw@gmail.com>

pkgname=acetoneiso2
pkgver=2.3
pkgrel=4
pkgdesc="An all in one ISO tool (bin mdf nrg img daa dmg cdi b5i bwi pdi iso)"
arch=('i686' 'x86_64')
url="http://www.acetoneteam.org"
license=('GPL3')
depends=('qtwebkit' 'fuseiso' 'cdrdao' 'cdrkit' 'dvd+rw-tools' 'gnupg2' 'pinentry' 'mplayer' 'phonon')
optdepends=("p7zip: for ISO compress/decompress support")
conflicts=('acetoneiso')
install=${pkgname}.install
source=(http://downloads.sourceforge.net/acetoneiso/AcetoneISO/AcetoneISO%202.3/acetoneiso_${pkgver}.tar.gz)
md5sums=('225f7af687016690ff28b9771a1ce179')

build() {
	cd ${srcdir}/acetoneiso_${pkgver}/acetoneiso

	qmake
	# needed for phonon
	sed -i 's|-I/usr/include/QtCore|-I/usr/include/QtCore -I/usr/include/KDE|' Makefile
	make
}

package() {
	cd ${srcdir}/acetoneiso_${pkgver}/acetoneiso

	make INSTALL_ROOT="${pkgdir}" install
}
